Ecuaciones Diferenciales
grupo 10 semestre 2022-2
Segundo Examen Parcial: Temas 3 & 4
SOLUCIÓN

2022-05-26

PREGUNTA 1 (20 puntos) Mediante la Transformada de Laplace obtenga la solución de la ecuación diferencial, sujeta a las condiciones iniciales dadas (***sin usar dsolve***)

> $Ecua := 4 \cdot diff(y(t), t\$2) + y(t) = \cos(t) \cdot \mathrm{Heaviside}(t-2\cdot \mathrm{Pi})$

$$Ecua := 4\left(\frac{d^2}{dt^2}\, y(t)\right) + y(t) = \cos(t)\ \mathrm{Heaviside}(t-2\pi) \quad (1)$$

> $Cond := y(0) = 0, \mathrm{D}(y)(0) = 1$

$$Cond := y(0) = 0, \mathrm{D}(y)(0) = 1 \quad (2)$$

> $with(inttrans):$

> $EcuaTransLap := subs(Cond, laplace(Ecua, t, s))$

$$EcuaTransLap := 4\, s^2\, laplace(y(t), t, s) - 4 + laplace(y(t), t, s) = \frac{e^{-2 s \pi} s}{s^2+1} \quad (3)$$

> $SolTransLap := isolate(EcuaTransLap, laplace(y(t), t, s))$

$$SolTransLap := laplace(y(t), t, s) = \frac{\frac{e^{-2 s \pi} s}{s^2+1} + 4}{4\, s^2 + 1} \quad (4)$$

> $SolPart := invlaplace(SolTransLap, s, t)$

$$SolPart := y(t) = 2 \sin\left(\frac{1}{2}\, t\right) - \frac{1}{3}\ \mathrm{Heaviside}(t-2\pi)\left(\cos(t) + \cos\left(\frac{1}{2}\, t\right)\right) \quad (5)$$

> $Comprob := simplify(eval(subs(y(t) = rhs(SolPart), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprob := 0 = 0 \quad (6)$$

> $plot(rhs(SolPart), t = 0..6\cdot \mathrm{Pi})$



> *restart*

PREGUNTA 2 (20 puntos) Obtener la solución particular del sistema de ecuaciones diferenciales con las condiciones iniciales dadas (***sin usar dsolve***)

> *SistEcua* := *diff*(x[1](t), t) = x[2](t), *diff*(x[2](t), t) = -4·x[1](t) + 2·sin(t) : *SistEcua*[1]; *SistEcua*[2]

$$\frac{d}{dt}\, x_1(t) = x_2(t)$$

$$\frac{d}{dt}\, x_2(t) = -4\, x_1(t) + 2\sin(t) \qquad \textbf{(7)}$$

> *Cond* := x[1](0) = 0, x[2](0) = 0

$$Cond := x_1(0) = 0, x_2(0) = 0 \qquad \textbf{(8)}$$

> *AA* := *array*([[0, 1], [-4, 0]])

$$AA := \begin{bmatrix} 0 & 1 \\ -4 & 0 \end{bmatrix} \qquad \textbf{(9)}$$

> *BB* := *array*([0, 2·sin(t)])

$$BB := \begin{bmatrix} 0 & 2\sin(t) \end{bmatrix} \qquad \textbf{(10)}$$

> *with*(*linalg*) :

> $MatExp := exponential(AA, t)$

$$MatExp := \begin{bmatrix} \cos(2\,t) & \frac{1}{2}\sin(2\,t) \\ -2\sin(2\,t) & \cos(2\,t) \end{bmatrix} \quad \textbf{(11)}$$

> $Xcero := array([0, 0])$

$$Xcero := \begin{bmatrix} 0 & 0 \end{bmatrix} \quad \textbf{(12)}$$

> $SolGral := evalm(MatExp \,\&*\, Xcero)$

$$SolGral := \begin{bmatrix} 0 & 0 \end{bmatrix} \quad \textbf{(13)}$$

> $MatExpTau := map(rcurry(eval, t='t - \mathrm{tau}'), MatExp)$

$$MatExpTau := \begin{bmatrix} \cos(2\,t-2\,\tau) & \frac{1}{2}\sin(2\,t-2\,\tau) \\ -2\sin(2\,t-2\,\tau) & \cos(2\,t-2\,\tau) \end{bmatrix} \quad \textbf{(14)}$$

> $BBtau := map(rcurry(eval, t='\mathrm{tau}'), BB)$

$$BBtau := \begin{bmatrix} 0 & 2\sin(\tau) \end{bmatrix} \quad \textbf{(15)}$$

> $AAtau := evalm(\ MatExpTau \,\&*\, BBtau)$

$$AAtau := \begin{bmatrix} \sin(2\,t-2\,\tau)\sin(\tau) & 2\cos(2\,t-2\,\tau)\sin(\tau) \end{bmatrix} \quad \textbf{(16)}$$

> $SolPart := map(int, AAtau, \mathrm{tau}=0..t) : x[1](t) = SolPart[1]; x[2](t) = SolPart[2]$

$$x_1(t) = -\frac{1}{3}\sin(2\,t) + \frac{2}{3}\sin(t)$$

$$x_2(t) = -\frac{2}{3}\cos(2\,t) + \frac{2}{3}\cos(t) \quad \textbf{(17)}$$

> $CompUno := eval(subs(x[1](t) = SolPart[1], x[2](t) = SolPart[2], lhs(SistEcua[1]) - rhs(SistEcua[1]) = 0)\ )$

$$CompUno := 0 = 0 \quad \textbf{(18)}$$

> $CompDos := eval(subs(x[1](t) = SolPart[1], x[2](t) = SolPart[2], lhs(SistEcua[2]) - rhs(SistEcua[2]) = 0)\ )$

$$CompDos := 0 = 0 \quad \textbf{(19)}$$

> $plot([SolPart[1], SolPart[2]], t=0..2\cdot\mathrm{Pi})$



> *restart*

PREGUNTA 3 (30 puntos) Detremine una solución completa de la ecuación diferencial utilizando el método de separación de variables para una constante de separación nula (***sin usar pdsolve***)

> *Ecua* := *diff*(*y*(*x*, *t*), *x*$2) + *diff*(*y*(*x*, *t*), *x*, *t*) = 4·*t*^3·*diff*(*y*(*x*, *t*), *x*)

$$Ecua := \frac{\partial^2}{\partial x^2}\, y(x, t) + \frac{\partial^2}{\partial x\, \partial t}\, y(x, t) = 4\, t^3 \left(\frac{\partial}{\partial x}\, y(x, t)\right) \qquad \textbf{(20)}$$

> *EcuaSeparable* := *eval*(*subs*(*y*(*x*, *t*) = *F*(*x*)·*G*(*t*), *Ecua*))

$$EcuaSeparable := \left(\frac{d^2}{dx^2}\, F(x)\right) G(t) + \left(\frac{d}{dx}\, F(x)\right)\left(\frac{d}{dt}\, G(t)\right) = 4\, t^3 \left(\frac{d}{dx}\, F(x)\right) G(t) \qquad \textbf{(21)}$$

> $$EcuaSeparada := \frac{\left(lhs(EcuaSeparable) - \left(\frac{d}{dx}\, F(x)\right)\left(\frac{d}{dt}\, G(t)\right)\right)}{\left(\frac{d}{dx}\, F(x)\right) G(t)} = simplify\left(\frac{\left(rhs(EcuaSeparable) - \left(\frac{d}{dx}\, F(x)\right)\left(\frac{d}{dt}\, G(t)\right)\right)}{\left(\frac{d}{dx}\, F(x)\right) G(t)}\right)$$

$$EcuaSeparada := \frac{\frac{d^2}{dx^2} F(x)}{\frac{d}{dx} F(x)} = \frac{4\, G(t)\, t^3 - \left(\frac{d}{dt} G(t)\right)}{G(t)} \quad (22)$$

> $EcuaX := lhs(EcuaSeparada) = 0$

$$EcuaX := \frac{\frac{d^2}{dx^2} F(x)}{\frac{d}{dx} F(x)} = 0 \quad (23)$$

> $EcuaT := rhs(EcuaSeparada) = 0$

$$EcuaT := \frac{4\, G(t)\, t^3 - \left(\frac{d}{dt} G(t)\right)}{G(t)} = 0 \quad (24)$$

> $SolX := dsolve(EcuaX)$

$$SolX := F(x) = \_C1\, x + \_C2 \quad (25)$$

> $SolT := dsolve(EcuaT)$

$$SolT := G(t) = \_C1\, e^{t^4} \quad (26)$$

> $SolGralCero := y(x, t) = rhs(SolX) \cdot subs(\_C1 = 1, rhs(SolT))$

$$SolGralCero := y(x, t) = (\_C1\, x + \_C2)\, e^{t^4} \quad (27)$$

> $Comprobacion := simplify(eval(subs(y(x, t) = rhs(SolGralCero), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprobacion := 0 = 0 \quad (28)$$

> $restart$

PREGUNTA 4 (30 puntos) Determinar la solución de la ecuación diferencial considerando una constante de separación positiva (***sin usar pdsolve***)

> $Ecua := diff(z(x, y), x, y\$2) = diff(z(x, y), y)$

$$Ecua := \frac{\partial^3}{\partial y^2\, \partial x} z(x, y) = \frac{\partial}{\partial y} z(x, y) \quad (29)$$

> $EcuaSeparable := eval(subs(z(x, y) = F(x) \cdot G(y), Ecua))$

$$EcuaSeparable := \left(\frac{d}{dx} F(x)\right)\left(\frac{d^2}{dy^2} G(y)\right) = F(x)\left(\frac{d}{dy} G(y)\right) \quad (30)$$

> $EcuaSeparada := \frac{lhs(EcuaSeparable)}{\left(\frac{d}{dx} F(x)\right) \cdot \left(\frac{d}{dy} G(y)\right)} = \frac{rhs(EcuaSeparable)}{\left(\frac{d}{dx} F(x)\right) \cdot \left(\frac{d}{dy} G(y)\right)}$

$$EcuaSeparada := \frac{\frac{d^2}{dy^2} G(y)}{\frac{d}{dy} G(y)} = \frac{F(x)}{\frac{d}{dx} F(x)} \quad (31)$$

> $EcuaX := rhs(EcuaSeparada) = \beta^2$

$$EcuaX := \frac{F(x)}{\frac{d}{dx} F(x)} = \beta^2 \quad (32)$$

> $EcuaY := lhs(EcuaSeparada) = \beta^2$

$$EcuaY := \frac{\frac{d^2}{dy^2} G(y)}{\frac{d}{dy} G(y)} = \beta^2 \quad (33)$$

> $SolX := dsolve(EcuaX)$

$$SolX := F(x) = \_C1 \, e^{\frac{x}{\beta^2}} \quad (34)$$

> $SolY := dsolve(EcuaY)$

$$SolY := G(y) = \_C1 + \_C2 \, e^{\beta^2 y} \quad (35)$$

> $SolGral := z(x, y) = subs(\_C1 = 1, rhs(SolX)) \cdot rhs(SolY)$

$$SolGral := z(x, y) = e^{\frac{x}{\beta^2}} \left(\_C1 + \_C2 \, e^{\beta^2 y}\right) \quad (36)$$

> $Comp := eval(subs(z(x, y) = rhs(SolGral), lhs(Ecua) - rhs(Ecua) = 0))$

$$Comp := 0 = 0 \quad (37)$$

>

FIN DE LA SOLUCIÓN

>